

5. ΤΕΛΕΣΤΕΣ

Σύμφωνα με το συντακτικό της PROLOG, όταν γράφουμε ένα κατηγορημα (ή ένα συναρτησιακό όρο) τοποθετούμε πρώτα το κατηγορηματικό (ή συναρτησιακό) σύμβολο και στη συνέχεια παραθέτουμε, μέσα σε παρένθεση, τα ορίσματα που αντιστοιχούν στο κατηγορημα (ή τον συναρτησιακό όρο). Πολλές φορές όμως είναι πιο κατάλληλο να τοποθετήσουμε το κατηγορηματικό (ή συναρτησιακό) σύμβολο ανάμεσα στα ορίσματα του. Ετσι για παράδειγμα θα ήταν καλύτερα να γράψουμε την έκφραση ¹:

$2 + 3$ αντί του όρου $+(2,3)$

Αυτό γίνεται δυνατόν, εάν ορίσουμε το σύμβολο αυτό σαν ένα ειδικό **τελεστή** (*operator*) με τη βοήθεια μηχανισμού που μας παρέχει η PROLOG.

Γενικά ένας τελεστής χαρακτηρίζεται από:

- (α) το ονομά του
- (β) τον τύπο του (τη θέση του δηλαδή σε σχέση με τα ορίσματα του)
- (γ) μία προτεραιότητα (*precedence*)

Το όνομα του τελεστή ταυτίζεται με το όνομα του συναρτησιακού (ή κατηγορηματικού) συμβόλου που θέλουμε να ορίσουμε.

Ενας τελεστής μπορεί να είναι τύπου:

¹ Θα πρέπει να επισημάνουμε ότι η έκφραση αυτή δεν αντιστοιχεί στον αριθμό 5, αλλά στη δομή

$$\begin{array}{c} + \\ / \quad \backslash \\ 2 \quad 3 \end{array}$$

- ενθεματικού (infix)
- προθεματικού (prefix)
- επιθεματικού (postfix)

Ενας ενθεματικός τελεστής τοποθετείται ανάμεσα σε δύο ορίσματα, ενώ ένας προθεματικός τελεστής προηγείται του ενός ορίσματος που μπορεί να έχει και ένας επιθεματικός τελεστής γράφεται μετά το όρισμά του.

Σε κάθε τελεστή αντιστοιχεί μία προτεραιότητα, που είναι ένας ακέραιος αριθμός, συνήθως από το 1 ως το 1200.² Όταν ο τελεστής βρίσκεται σε μία έκφραση που υπάρχουν και άλλοι τελεστές, η προτεραιότητα αυτή χρησιμοποιείται, για να ξεκαθαριστεί η δομή της έκφρασης σύμφωνα με τον εξής κανόνα:

Κανόνας Προτεραιοτήτων: Ο τελεστής με την υψηλότερη προτεραιότητα σε ένα σύνθετο όρο αποτελεί το βασικό συναρτησιακό σύμβολο.

Εστω για παράδειγμα η έκφραση

$$a+b/c$$

Εάν ο τελεστής "+" έχει υψηλότερη προτεραιότητα από τον τελεστή "/", τότε η παραπάνω έκφραση είναι ισοδύναμη με τον όρο:

$$+(a, /(b,c))$$

Παρατηρήστε ότι με δεδομένες τις παραπάνω προτεραιότητες των "+" και "/", εάν θέλουμε να παραστήσουμε τον όρο

$$/((+a,b),c)$$

² Ο αριθμός αυτός πιθανά διαφέρει σε κάποιες υλοποιήσεις της PROLOG

σε ενθεματική μορφή, πρέπει να χρησιμοποιήσουμε υποχρεωτικά παρενθέσεις (Όταν μία έκφραση περικλείεται σε παρενθέσεις, θεωρούμε ότι έχει μηδενική προτεραιότητα):

$$(a+b)/c$$

Εάν στον ίδιο όρο υπάρχουν δύο τελεστές με την ίδια προτεραιότητα, τότε το ποιός θα θεωρηθεί σαν πιο βασικός καθορίζεται από την προσεταιριστικότητα τους. Η προσεταιριστικότητα συνεπάγεται από τον τρόπο ορισμού του τύπου ενός τελεστή.

Οι πιθανοί τύποι ενός ενθεματικού (*infix*) τελεστή είναι

$$x f x, \quad x f y \quad \text{και} \quad y f x$$

Όταν ένας τελεστής είναι τύπου $x f x$, πρέπει οι υποεκφράσεις που αποτελούν τα ορίσματα του τελεστή να περιέχουν υποχρεωτικά τελεστές μικρότερης προτεραιότητας απ' ότι ο ίδιος ο τελεστής. (Εκτός φυσικά εάν υπάρχουν παρενθέσεις.) Ένας τελεστής τύπου $x f x$ ονομάζεται **μη προσεταιριστικός** (*non associative*).

Όταν ένας τελεστής είναι τύπου $x f y$, τότε οι τελεστές που βρίσκονται στην αριστερή υποέκφραση πρέπει να είναι υποχρεωτικά χαμηλότερης προτεραιότητας, ενώ οι τελεστές που βρίσκονται στην δεξιά υποέκφραση μπορούν να είναι και της ίδιας προτεραιότητας. Ένας τελεστής τύπου $x f y$, λέμε ότι έχει προσεταιριστικότητα από τα δεξιά προς τα αριστερά.

Παράδειγμα 5.1

Εστω η έκφραση:

$$a+b-c$$

(I)

Εάν οι τελεστές "+" και "-" οριστούν με την ίδια προτεραιότητα και τύπο yfx , τότε η (I) είναι ισοδύναμη με τον όρο:

$$-(+(a,b),c)$$

Εάν οι τελεστές "+" και "-" οριστούν με τύπο xfy , τότε η (I) είναι ισοδύναμη με τον όρο:

$$-(a,+(b,c))$$

Εάν όμως οι τελεστές "+" και "-" οριστούν με τύπο xfx , τότε η (I) είναι συντακτικά λανθασμένη.

Οι πιθανοί τύποι ενός προθεματικού (*prefix*) τελεστή μπορεί να είναι:

$$fx \text{ και } fy$$

και ενός επιθεματικού (*postfix*):

$$xf \text{ και } yf$$

Η σημασία τους συνάγεται κατ' αναλογία από τη σημασία των τύπων του ενθεματικού τελεστή.

Παράδειγμα 5.2:

Εάν το σύμβολο **not** οριστεί σαν ένας προθεματικός τελεστής τύπου fy τότε η έκφραση:

$$\text{not not } P$$

(II)

είναι αποδεκτή και αντιστοιχεί στον όρο:

not(not(P))

Εάν όμως ο τύπος του **not** οριστεί σαν **fx** τότε η (II) είναι συντακτικά λανθασμένη.

Στον πίνακα 5.1 συνοψίζονται οι διαφορετικοί τύποι που μπορεί να πάρει ένας τελεστής.

Ο ορισμός ενός τελεστή με όνομα **Name**, τύπο **Type** και προτεραιότητα **Precedence**, μπορεί να γίνει με την εκτέλεση του ενσωματωμένου κατηγορήματος **op**:

?- op(Precedence, Type, Name).

Στη θέση του **Name** μπορούμε επίσης να έχουμε και μία λίστα από ονόματα, στην περίπτωση που θέλουμε να ορίσουμε περισσότερους από έναν τελεστές που έχουν τον ίδιο τύπο και προτεραιότητα.

Θέση	Τύπος	Προσεταιριστικότητα
Infix	xfx	Μη-Προσεταιριστικός
	xfy	Από Δεξιά προς Αριστερά
	yfx	Από Αριστερά προς Δεξιά
Prefix	fx	Μη-Προσεταιριστικός
	fy	Από Δεξιά προς Αριστερά
Postfix	xf	Μη-Προσεταιριστικός
	yf	Από Αριστερά προς Δεξιά

Πίνακας 5.1 : Οι τύποι ενός τελεστή (*operator*)

Τελεστής	Τύπος	Προτεραιότητα	
		(Edinburg PROLOG)	(Iso Standard PROLOG)
mod	xfx	11	400
*	yfx	21	400
/	yfx	21	400
+	fx	--(*)	500
-	fx	--(*)	500
+	yfx	31	500
-	yfx	31	500
=	xfx	40	700
==	xfx	40	700
\=	xfx	40	700
\==	xfx	40	700
<	xfx	40	700
>	xfx	40	700
=<	xfx	40	700
>=	xfx	40	700
is	xfx	40	700
=..	xfx	40	700
not	fy	60	900
,	xfy	253	1000
;	xfy	254	1100
?-	fx	255	1200
:-	fx	--(*)	1200
:-	xfx	255	1200

Πίνακας 5.2 : Συνήθεις προκαθορισμένοι τελεστές

(*) Στην Edinburg PROLOG οι τελεστές αυτοί χρησιμοποιούνται μόνο σαν ενθεματικοί

```

derivative(X,X,1) :- !.
derivative(C,X,0) :- C \==X.
derivative(sin(X),X,cos(X)).
derivative(cos(X),X,-sin(X)).
derivative(U+V,X,A+B) :-
    derivative(U,X,A),
    derivative(V,X,B).
derivative(U*V,X,A*V+B*U) :-
    derivative(U,X,A),
    derivative(V,X,B).

?- derivative(x*x+2,x,A).

```

πρόγραμμα 5.1: Συμβολική παραγωγή

Κάθε σύστημα της PROLOG διαθέτει ένα σύνολο από προκαθορισμένους τελεστές. Ένα τυπικό σύνολο τέτοιων τελεστών δίνεται στον πίνακα 5.2. Δίνονται επίσης οι προτεραιότητες των τελεστών αυτών στην κατά ISO Standard PROLOG (Από 1 μέχρι 1200) και - για ιστορικούς λόγους - στην Edinburgh PROLOG (Ένας αριθμός από 1 μέχρι 255).

Η δυνατότητα ορισμού τελεστών αποτελεί βασικό πλεονέκτημα για την PROLOG, γιατί ο χρήστης μπορεί να προσαρμόσει το συντακτικό της γλώσσας στις ανάγκες του. Έτσι για παράδειγμα ένα κατηγορημα της μορφής:

```
is_brother_of(john,nick).
```

μπορεί να γραφεί πιο ευανάγνωστα σαν

```
john is_brother_of nick.
```

εάν το κατηγορηματικό σύμβολο `is_brother_of` οριστεί σαν ενθεματικός (*infix*) τελεστής με την

?- op(670, xfx, is_brother_of).

Ενα παράδειγμα προγράμματος, που αφορά στη συμβολική παραγωγή όπου φαίνεται η χρήση των τελεστών "+" και "*", δίνεται από το πρόγραμμα 5.1.

Αξίζει να σημειωθεί η δηλωτική φύση του προγράμματος αυτού. Εκτός από συντακτικές διαφορές οι φράσεις που ορίζουν την παράγωγο αντιστοιχούν πλήρως στο μαθηματικό ορισμό της παραγώγου. Η απάντηση στην ερώτηση είναι ($A=x*1+x*1+0$).